The goal of this illustration is to develop your intuition about the three dimensional figures.
Use this code as a **guide** and visualize each figure in your head.

To access the code Type mupad on MATLAB command prompt or go to
start->Toolboxes-> symbolic math-> mupad

The following figure is for Tutorial 6 problem 3. Cylinder equation $x^2+y^2=2ay$, a=1.

```
h:=plot::Cylindrical([2*sin(phi), phi, r],r=0..2, phi = 0..PI/2,LineColor = RGB::Blue)
```

$$\text{plot::Cylindrical}\left([2\,\sin(\text{phi}),\ \text{phi},\ r],\ r = 0..2,\ \text{phi} = 0..\frac{\pi}{2}\right)$$

Plot of sphere $x^2+y^2+z^2=4a^2$, a=1

```
h0:=plot::Spherical([2, u, v], u = 0..PI/2, v = 0..PI/2,LineColor = RGB::Red)
```

$$\text{plot::Spherical}\left([2,\ u,\ v],\ u = 0..\frac{\pi}{2},\ v = 0..\frac{\pi}{2}\right)$$

Curve of intersection by letting y=t, solving for x in terms of t from equation $x^2+y^2=2ay$, a=1
and z in terms of t from equation $x^2+y^2+z^2=4a^2$, a=1

```
c0:=plot::Curve3d([sqrt(2*t-t^2),t,sqrt(4-2*t)], t = 0..2,LineColor = RGB::Green,LineWidth =2)
```

$$\text{plot::Curve3d}\left(\left[\sqrt{2\,t-t^2},\ t,\ \sqrt{2}\,\sqrt{-t+2}\right],\ t = 0..2\right)$$

```
plot(h,h0,c0)
```



The following plot is for Assignment 6 question 3. The plot of $x^2+y^2=a^2$, a=1

```
h1:=plot::Implicit3d(x^2+y^2-1,x=0..2,y=0..2,z=0..2,LineColor = RGB::Green)
```

$$\text{plot::Implicit3d}\left(x^2+y^2-1,\ x = 0..2,\ y = 0..2,\ z = 0..2\right)$$

The plot of $x^2+z^2=a^2$, a=1

```
h2:=plot::Implicit3d(x^2+z^2-1,x=0..2,y=0..2,z=0..2)
```

$$\text{plot::Implicit3d}\left(x^2+z^2-1,\ x = 0..2,\ y = 0..2,\ z = 0..2\right)$$

Curve of intersection, by letting x=t, and solving for y,z using the two equations $x^2+y^2=a^2$, a=1, $x^2+z^2=a^2$, a=1

```
c2:=plot::Curve3d([t,sqrt(1-t^2),sqrt(1-t^2)], t = 0..1,LineColor = RGB::Yellow,LineWidth =2)
```

$$\text{plot::Curve3d}\left(\left[t,\ \sqrt{-t^2+1},\ \sqrt{-t^2+1}\right],\ t = 0..1\right)$$

The following two plots are plots in xz and xy plane to show that there are two surfaces and symmetric.

```
c2xz:=plot::Curve3d([t,0,sqrt(1-t^2)], t = 0..1,LineColor = RGB::OliveGreen,LineWidth =2)
```

$$\text{plot::Curve3d}\left(\left[t,\ 0,\ \sqrt{-t^2+1}\right],\ t = 0..1\right)$$

```
c2xy:=plot::Curve3d([t,sqrt(1-t^2),0], t = 0..1,LineColor = RGB::PowderBlue,LineWidth =2)
```

$$\text{plot::Curve3d}\left(\left[t,\ \sqrt{-t^2+1},\ 0\right],\ t = 0..1\right)$$

```
plot(h2,h1,c2,c2xz,c2xy)
```



Surface of revolution. The following animation illustrates the idea of surface of revolution.
I had selected the simple function that is straight line. You can play around by using different function. You can change function at `f := u -> u:,` For exmaple to u^2.

This animation illustrates rotation along y direction. For viewing the animation click on the figure. In the top menu you can find the animationcontrol menu after clicking on the figure.

```
f := u -> u:
plot(plot::Surface([u, f(u)*sin(v), f(u)], u = 0 .. 2,
                   v = 0 ..a*2*PI, a = 0 .. 1),AnimationStyle = Loop):
```



Along Z direction

```
f := u -> u:
plot(plot::Surface([u, f(u), f(u)*cos(v)], u = 0 .. 2,
                   v = 0 ..a*2*PI, a = 0 .. 1),AnimationStyle = Loop):
```



Now both y and z. That is **surface of revolution**.

```
f :=  u -> u:
plot(plot::Surface([u, f(u)*sin(v), f(u)*cos(v)], u = 0 .. 2,
                    v = 0 ..a*2*PI, a = 0 .. 1),AnimationStyle = Loop):
```



The following plot is for assignment 6 question 5c. Using our usual steps as outlined in earlier plots.

```
h3:=plot::Function3d(x^2,x=0..1,y=0..1,z=0..1,LineColor = RGB::Green)
```

$$\text{plot::Function3d}\left(x^2,\ x = 0..1,\ y = 0..1\right)$$

```
h4:=plot::Implicit3d(z-1+3*x^2+y^2,x=0..2,y=0..2,z=0..2)
```

$$\text{plot::Implicit3d}\left(z + 3\,x^2 + y^2 - 1,\ x = 0..2,\ y = 0..2,\ z = 0..2\right)$$

```
c3:=plot::Curve3d([t,sqrt(1-4*t^2),t^2], t = 0..4,LineColor = RGB::Yellow,LineWidth =2)
```

$$\text{plot::Curve3d}\left(\left[t,\ \sqrt{-4\,t^2 + 1},\ t^2\right],\ t = 0..4\right)$$

```
plot(h4,h3,c3)
```



This animation is for Torus. Notice that we start from circle and roatate it along x axis. A nice tutorial video about torus
can be found on web, www.khanacademy.org.

```
plot(plot::Surface([(5+2*cos(v))*sin(u), (5+2*cos(v))*cos(u), 2*sin(v)], u = 0 .. a*2*PI,
                    v = 0 .. 2*PI, a = 0 .. 1),AnimationStyle = Loop):
```


2
1
z
0
-1
-2
6
4
2
0
-2
-4
-6
y
-6
-4
-2
0
2
4
6
x